
13. CASE STUDY I: ENGLISH PHRASE GENERATOR  

We start our case studies with a simple one. The object of this program is to generate phrases based on a subset 
of English. (You can substitute the rules for any language later on.)  

We use an approach called context-free grammar (CFG) – a grammar used to describe a language without 
reference to its context. It should be noted however that while CFGs simplify the procedure of phrase 
generation, they usually render the output lacking in “true intelligence”.  

Here’s an example of a CFG for a subset of English:  

Sentence ⇒ Noun Phrase + Verb Phrase  

Noun Phrase ⇒ Article + Noun  

Verb Phrase ⇒ Verb + Noun Phrase  

Article ⇒ the, an, a etc  

Noun ⇒ man, woman, ball, table etc  

Verb ⇒ hit, took, saw, liked etc.  

Notice that this is the way linguists usually describe CFGs. This will play a role in our discussion later.  

13.1. The obvious approach  

We can see from the rules that we can simply append a verb phrase and a noun phrase to get a sentence. The 
obvious approach for converting these rules to a lisp program would be as follows:  

(defun sentence ()     (append (noun-phrase) (verb-phrase))) 
(defun noun-phrase()   (append (Article) (Noun))) 
(defun verb-phrase()   (append (Verb) (Noun-phrase))) 
(defun Article()       (one-of '(the a))) 
(defun Noun()          (one-of '(man ball woman table))) 
(defun Verb()          (one-of '(hit took saw liked))) 

one-of is a simple function which picks one element at random from the list provided to it. (Look up the built-
in function elt in a reference.) It returns a list of on element. This is because we would want to append this 
element later.  

(defun one-of (set) 
  (list (random-elt set))) 
 
(defun random-elt (choices) 
  (elt choices (random (length choices)))) 

Problems with the obvious approach 

Remember the purpose of Artificial Intelligence programs. They need to be extensible before anything else. If a 
program is written easily and is efficient but is difficult to extend or use in a practical environment, it is not a 
good AI program. The main problem with the above approach is that it relies on the user’s knowledge of lisp 
syntax. What would happen, for example, if the user decided to extend the grammar by adding new rules?  

The real reason for using lisp instead of another language for AI programs is that you can define new languages 
in lisp; languages which the users understand. Lisp programs convert these languages to lisp syntax on-the-fly.  

This leads to another approach for tackling this problem.  

13.2. A rule-based approach 

We want to be able to write a program in which we can define the rules in the following syntax.  

(defparameter *simple-grammar*  
  '((sentence      ->  (noun-phrase verb-phrase)) 
    (noun-phrase   ->  (article noun)) 
    (verb-phrase   ->  (verb noun-phrase)) 
    (article       ->  the a) 
    (noun          ->  man ball woman table) 
    (verb          ->  hit took saw liked)) 
  "A simple grammar for english sentences") 

As you can see, this language is easier to understand for the user as it’s closer to the original problem (complete 
with the arrow and all). We could make it look exactly as the original problem using macros but we want to keep 
the solution simple right now.  

We define a variable to hold this grammar:  

(defvar *grammar* *simple-grammar*) 

Now, the solution is a little more complex. We have the rules but we don’t have a program which understands 
them. Let’s write one.  

First we define functions for computing the left and right hand side of any given rule. (Please note that the arrow 
in the rules is purely decorative. It serves no real purpose for our program.)  

(defun rule-lhs (rule) 
  (first rule)) 
(defun rule-rhs (rule) 
  (cdr (cdr rule))) 

Now we need to write a function which calculates all rewrites for any given category. For example, a 
rewrite for sentence is noun-phrase + verb phrase (the parentheses around the two mean that both must 
be appended) and a possible rewrite for a noun is ball. Here’s the function:  

(defun rewrites (category) 
  "Returns a list of possible rewrites for this category" 
  (rule-rhs (assoc category *grammar* ))) 

And finally, we write the heart of our generator:  

(defun generate (phrase) 
  "Generate a random phrase" 
  (cond ((listp phrase)     ;; case 3 
         (mappend #'generate phrase)) 
        ((rewrites phrase)    ;; case 2 
         (generate (random-elt (rewrites phrase)))) 
        (t (list phrase))))    ;; case 1  

Looks intimidating right now but it’s really quite simple.  

The cond has three cases. We’ll discuss them in reverse order.  

Case 1. The phrase passed is not a list and does not have any rewrites. This simply means that there is no 
rule with the “phrase” as its left-hand side. An example of this is ‘ball’. Phrase ‘noun’ does have rewrites 
so it does not fall in this category.   

Case 2. The phrase passed is not a list but it does have some rewrites. For example, ‘noun’. For this kind 
of phrases, we pick a random possible rewrite and generate that. In our example of ‘noun’ that could be the 
phrase ‘man’. We pick man (at random) and call generate with ‘man’ as phrase. This would become case 1 
and would return (man).  

Case 3. If the phrase is a list – for example, (noun-phrase verb-phrase), we “mappend” 
generate on the list. mappend simply applies the function on all elements of the list and appends the result.  

(defun mappend (fn set) 
  "Apply fn to all elements of set and append the results" 
  (apply #'append (mapcar fn set))) 

random-elt has been defined earlier. We’ll be using it often in the future.  

We’re doing all this appending because our function ‘generate’ returns a single word as a list with one 
element (case 1). We want to append all these results in one big list to get rid of all the parentheses.  

Remember this approach of trying to understand lisp programs. In the future case studies, I won’t be giving this 
thorough an explanation. 


